Cahiers GUTenberg

MACROTEX : UN GÉNÉRATEUR DE CODE LaTeX IMPLÉMENTÉ EN MACSYMA

J. Ph. Chancelier, A. Sulem

Cahiers GUTenberg, nº 3 (1989), p. 32-39.

<http://cahiers.gutenberg.eu.org/fitem?id=CG_1989___3_32_0>

MACROTEX : un générateur de code LATEX implémenté en MACSYMA

J. Ph. CHANCELIER et A. SULEM

INRIA, Domaine de Voluceau, Rocquencourt, BP 105, 78153 Le Chesnay Cédex, France

Abstract

MACROTEX is a code generator producing LATEX code from Macsyma, a system for symbolic calculus. MACROTEX contains both translation functions from Macsyma into LATEX and paper generation facilities. So it adds the power of Macsyma to the quality of texts edited through LATEX. MACROTEX is written in Common-Lisp.

1. Introduction

Macsyma [4] est un langage de calcul formel conçu pour faire des calculs numériques et des manipulations algébriques (dérivation, intégration, résolution d'équations, simplification d'expressions, calcul matriciel ...).

LATEX [1] est un programme permettant d'obtenir une grande qualité d'édition, en particulier d'édition scientifique. (Rappelons que LATEX rajoute à TEX une collection de commandes permettant de simplifier l'édition de texte).

Notre but est de réunir Macsyma et LATEX afin d'obtenir un outil facile à utiliser associant la puissance de manipulation algébrique de Macsyma et la qualité d'édition de LATEX. Un utilisateur peut ainsi faire du calcul algébrique dans Macsyma et générer du code LATEX sans avoir à connaître la syntaxe LATEX. Nous avons donc écrit un générateur de code LATEX en Common Lisp que nous avons appelé "MACROTEX". Ce programme tourne sur Machine Symbolics Lisp, release-7 en Common Lisp [5] et sur Sun en Franz Lisp.

MACROTEX comprend un ensemble d'instructions pour la traduction d'expressions mathématiques Macsyma en syntaxe LATEX (en particulier une instruction permettant de découper les grosses formules automatiquement). De plus, des facilités sont rajoutées pour générer un document LATEX au niveau Macsyma.

2. Description de MACROTEX

2.1. Le traducteur Macsyma-LATEX

On décrit ici les fonctions de base du traducteur illustrées d'exemples (on donnera l'instruction MACROTEX suivie du code LATEX généré). Une documentation plus complète avec un manuel de référence se trouve en [2].

- La fonction de base TEX(*exp*) a pour argument une expression mathématique en syntaxe Macsyma qu'elle traduit en syntaxe LATEX et facultativement une étiquette pour la numéroter.

```
(C1) TEX(SIN(X)+X/Y-INTEGRATE
          (F(X),X,0,1),LABEL);
```

$$\frac{X}{Y} - \int_0^1 F(X)\,dX + \sin(X) \qquad (1)$$

- La fonction EXPTXT(*exp*) est analogue à la fonction TEX pour une formule à insérer dans du texte.
- La fonction FORMATT(*string, args*) est similaire à la fonction Lisp *format* [5], avec les directives de formattage supplémentaires suivantes :
  - ~K and ~L traduisent des formules en LATEX utilisant respectivement les fonctions TEX et EXPTXT.

•~N traduit des expressions non mathématiques en LaTeX.

•~Y et ~Z permettent d'insérer respectivement des références d'équations ou des références bibliographiques.

```
(C3) FORMATT(" Exemple de ~N
          dans du texte :
          ~L.","formule",
          SUM(X(I),I,A,B));
```

Exemple de formule dans du texte : $\sum_{I=A}^{B} X(I)$.

• La fonction LISTEXPT a pour argument une liste d'expressions mathématiques et facultativement un argument précisant l'alignement et une étiquette. Elle écrit le système d'expressions l'une en dessous de l'autre en syntaxe LaTeX selon l'alignement souhaité : centré (par défaut), rectifié à droite ou à gauche.

```
(C5) LISTEXPT([SUM(X(I)^N3,I,
          N1,N2),
          INTEGRATE(F(X),
          X,0,INF)]);
```

$$\sum_{I=N1}^{N2} X(I)^{N3}$$

$$\int_0^{+\infty} F(X)\, dX$$

• La fonction DECOUPET permet de découper une longue formule.

```
(C6) DECOUPET(F(X),
        [COS(Y),1.23*
        INTEGRATE(G(X),X,0,1),
        SIN(X)],["+","-"]);
```

$$\begin{array}{rcl} F(X) & = & \cos(Y) \\ & + & 1.23 \int_0^1 G(X)\, dX \\ & - & \sin(X) \end{array}$$

• La macro-instruction DECOUPEM s'utilise quand on ne sait pas a priori comment découper une longue formule. Elle a pour premier argument une formule Macsyma qu'elle traduit en Latex après l'avoir découpée en sous-formules dont la taille est contrôlée par la valeur d'un deuxième argument (l'étiquette est optionnelle).

```
(C7) DECOUPEM(MATRIX
       ([SUM(SUM(X[I,J],I,1,N1),
        J,1,N2),
        INTEGRATE(F(X),X,0,1)],
        [DIFF(F(X),X[1],N),
        PRODUCT(X[I],I,1,N)]),
        LAB,60);
```

L'utilisateur peut lui-même définir des fonctions de traduction particulières d'opérateurs Macsyma.

$$\begin{gathered} \begin{pmatrix} E1 & E2 \\ \frac{d^N F(X)}{dX^N} & \prod_{I=1}^{N} X_I \end{pmatrix} \\ E1 = \sum_{J=1}^{N2} \sum_{I=1}^{N1} X_{I\,J} \\ E2 = \int_0^1 F(X)\, dX \end{gathered} \tag{2}$$

2.2. Le générateur de documents LaTeX

2.2.1. Les instructions élémentaires

Ce sont les fonctions du traducteur Macsyma-LaTeX (TEX, EXPTXT, FORMATT, ...) et des fonctions Macsyma qui génèrent une instruction LaTeX particulière telles que décrites dans le tableau 1. On a également défini des instructions de mise à jour de listes, comme ADD_REFERENCET et ADD_NOTATIONT qui permettent de stocker dans des piles les références et les notations utiles pour la génération d'un document. L'instruction CONTEXTT initialise ces piles et précise les paramètres de style du document.

2.2.2. Les macro-instructions

Les macro-instructions sont des commandes qui génèrent plusieurs instruc-

Tableau 1 : Fonctions du traducteur

MACROTEX	LATEX
documentstylet("article"11) labelt ("Euler") reft ("Newton") citet ("aut1") sectiont (" méthode de ~N","résolution")	\documentstyle[11pt]{article} \label{Euler} \ref{Newton} \cite{aut1} \section[méthode de résolution] { méthode de résolution}

Tableau 2 : Quelques macro-instructions

MACROTEX	LATEX
abstractm(prog)	\begin{abstract} prog \end{abstract}
itemizem (item1 ...)	\begin{itemize} \item item1 ... \end{itemize}
stylem (doc,font,height,width,top,odd, even,par)	\documentstyle[font]{doc} \textheight= height pt \textwidth=width pt \topmargin=top pt \oddsidemargin=odd pt \evensidemargin=even pt \marginparwidth=par pt
citem([key1,source1], ...)	\cite[key1] et met à jour la pile des références
notationm()	\section{notation} \begin{itemize} notation1 ... \end{itemize}
bibliom()	\begin{thebibliography}{3} \bibitem{key1} source1 ... \end{thebibliography}
reportm(titre,auteurs,résumé,chapitres)	style titre auteurs \begin{document} \maketitle \tableofcontents résumé notations chapitres références \end{document}

tions élémentaires. Le tableau 2 en montre quelques-unes.

2.2.3. *Génération du code LATEX*

Un programme MACROTEX est construit en programmant dans Macsyma. C'est une liste Macsyma composée d'instructions MACROTEX (instructions élémentaires ou macro-instructions). Chaque instruction est une liste Macsyma dont le premier élément est une fonction MACROTEX et les suivants sont ses arguments. La traduction en LATEX est faite par l'appel de la fonction EXECUTE sur cette liste Macsyma. Par exemple, $execute([tex, a + b])$ sera équivalent à $tex(a + b)$. Il est possible d'envoyer directement le code généré dans un fichier.

3. Exemple de rapport généré

L'exemple suivant montre l'utilisation de MACROTEX pour générer un article.
On définit pour cela la fonction Macsyma décrite dans la figure 1.

Cette fonction utilise les constantes et les fonctions Macsyma montrée dans la figure 2.

L'appel de la fonction *control*(2, 2, v, [1,1], [u[1], u[2]], $u[1]^2 + u[2]^2$, "stochastique") génère l'article LATEX se trouvant en fin de cet article.

Références bibliographiques

[1] L. LAMPORT. LATEX. *A Document Preparation System.* Addison-Wesley Publishing Company, 1986.

[2] J.P. CHANCELIER - A.SULEM. *MACROTEX : A LATEX CODE GENERATOR IN MACSYMA.* Rapport technique Inria n°93 (décembre 1987).

[3] J.P. CHANCELIER - A.SULEM. "MACROTEX : A LATEX CODE GENERATOR IN MACSYMA", *MACSYMA newsletter*, Volume 5, number 3, July 1988.

[4] *MACSYMA Reference Manual.* Version 12 (1986). Symbolics, Inc.

[5] G. STEELE. *COMMON LISP. The language.* Digital Press.

```
control(n1,n2,v,df,dr,cf,alea):=block(
CONTEXTT(),
ADD_REFERENCET([Fleming,"FLEMING,W.H.-RISHEL,R.(1975).
               Deterministic and Stochastic Optimal Control.
               Springer Verlag, New York."]),
ADD_NOTATIONT([FORMATT," variables d'\\'etat: ~L ", xx],
              [FORMATT," variables de contr\\^ole: ~L ",uu],
execute(
[REPORTM ,
  [TITLET,  "Un  probl\\'eme de contr\\^ole  ~N  ", alea],
  [AUTHORT,["J.P. Chancelier-A.Sulem", "Inria"]],
  [ABSTRACTM, [FORMATT, "On consid\\'ere un probl\\'eme de
                         contr\\^ole ~N que l'on r\\'esout
                         par la programmation dynamique. ", alea ]],
  [[SECTIONT, "M\\'ethode de la programmation dynamique"],
     [FORMATT, "Le co\\^ut optimal satisfait l'\\'equation
               de Bellman ~Y avec des conditions
               dirichlet homog\\'ene.",Fleming],
     [TEX, minim(uu,auv+cf)-v = 0, eq],
     [FORMATT, "avec ~K", hamiltonien(df,dr,n1,v)],
     [SUBSECTIONT, "Discr\\'etisation:"],
        [FORMATT,"On note ~L le i-\\'eme pas de discr\\'etisation.
                 On d\\'efinit:",\h[\i]],
        [LISTEXPT, translation(n1,xx,v)] ,
        [FORMATT, "L'\\'equation de Bellman discr\\'etis\\'ee
                  est obtenue en rempla\\c{c}ant
                  l'op\\'erateur $A$ par ~L dans l'\\'equation ~Z
                  avec", A[\h], eq],
        [DECOUPEM, funmake(A[\h],uu).v = factorise(df,dr,n1,v)],
        [FORMATT, "Cela s'interpr\\'ete comme un probl\\'eme de
                  contr\\^ole de cha\\^{\\i}ne de Markov.
                  La matrice de transition est:~K",
                  matrice(df,dr,n1,xx,v)]]]))$
```

Figure 1 : Fonction Macsyma utilisée dans l'exemple

```
xx:makelist(\x[i],i,1,n1),
uu:makelist(\u[i],i,1,n2),
auv:funmake('A,uu).v,
translation(n,xx,v):=makelist(s[i](v)
       =funmake(v,subst(xx[i]+\h[i],xx[i],xx)),i,1,n)$
hamiltonien(aa,b,n,v):= auv=sum(aa[i].(D[i]^2).v+b[i].D[i].v,i,1,n)$
discret(a,b,n,v):= expand(sum(a[i]*d2(i,v)+b[i]*d1(i,v),i,1,n))$
factorise(a,b,n,v):= apply("+",map(lambda([x],
                          coeff(discret(a,b,n,v),x).x),
                          append([v],makelist(s[i].v,i,1,n),
                          makelist(s[i]^^(-1).v,i,1,n))))$
matrice(a,b,n,xx,v):= apply(matrix, append(
        [["Pt-Initial","Pt-Final","Probabilit\\'e-transition"],
        [xx,xx,0]],
        makelist(lign(discret(a,b,n,v),i,\h[i],
                 s[i].v,n,xx,v),i,1,n),
        makelist(lign(discret(a,b,n,v),i,-\h[i],
                 s[i]^^(-1).v,n,xx,v),i,1,n)))$
lign(l,i,s1,s2,n,xx,v):=[xx, subst(xx[i]+s1,xx[i],xx),
                        ratsimp(coeff(l,s2)/-coeff(l,v))]$
```

Figure 2 : Constantes et fonctions Macsyma utilisées par l'exemple

Un problème de contrôle stochastique

J.P. Chancelier-A. Sulem
Inria

Résumé

On considère un problème de contrôle stochastique que l'on résout par la programmation dynamique.

1. Notations

- Variables d'état: $[x_1, x_2]$
- Variables de contrôle: $[u_1, u_2]$

2. Méthode de la programmation dynamique

Le coût optimal V satisfait l'équation de Bellman [1] avec des conditions de Dirichlet homogène.

$$\min_{u_1,u_2} \left(A(u_1,u_2)V + u_2^2 + u_1^2\right) - V = 0 \tag{1}$$

avec

$$A(u_1,u_2)V = u_2 D_2 V + D_2^2 V + u_1 D_1 V + D_1^2 V$$

2.1. Discrétisation

On note h_i le i-ème pas de discrétisation. On définit:

$$S_1(V) = V(x_1 + h_1, x_2)$$

$$S_2(V) = V(x_1, x_2 + h_2)$$

L'équation de Bellman discrétisée s'obtient en remplaçant l'opérateur A par A_h dans l'équation (1) avec

$$\begin{aligned}
A_h(u_1,u_2)V &= E5 + E4 + E3 + E2 + E1 \\
E1 &= \left(\frac{0.5u_2}{h_2} + h_2^{-2}\right) S_2 V \\
E2 &= \left(h_2^{-2} - \frac{0.5u_2}{h_2}\right) S_2^{-1} V \\
E3 &= \left(-2h_1^{-2} - 2h_2^{-2}\right) V \\
E4 &= \left(\frac{0.5u_1}{h_1} + h_1^{-2}\right) S_1 V \\
E5 &= \left(h_1^{-2} - \frac{0.5u_1}{h_1}\right) S_1^{-1} V
\end{aligned}$$

Cela s'interprète comme un problème de contrôle de chaîne de Markov. La matrice de transition est:

$$\begin{pmatrix} Pt-Initial & Pt-Final & Probabilité-transition \\ [x_1,x_2] & [x_1,x_2] & 0 \\ [x_1,x_2] & [x_1+h_1,x_2] & \frac{(h_1u_1+2)h_2^2}{(4h_2^2+4h_1^2)} \\ [x_1,x_2] & [x_1,x_2+h_2] & \frac{(h_1^2h_2u_2+2h_1^2)}{(4h_2^2+4h_1^2)} \\ [x_1,x_2] & [x_1-h_1,x_2] & -\frac{(h_1u_1-2)h_2^2}{(4h_2^2+4h_1^2)} \\ [x_1,x_2] & [x_1,x_2-h_2] & -\frac{(h_1^2h_2u_2-2h_1^2)}{(4h_2^2+4h_1^2)} \end{pmatrix}$$

Références bibliographiques

[1] FLEMING,W.H.-RISHEL,R.(1975). Deterministic and Stochastic Optimal Control. Springer Verlag, New York.